
Part 1: Arrays 
Arrays are tabular-style variables, containing rows and columns. Web application developers 

often use arrays to store temporary data, such as shopping cart contents, without committing 

the information to a database.  An array can store multiple values under a single variable 

name. These values can be simple values or complex data types and are indexed by 

numerical position. Here is a simple array. 

aSelection[1] = "Roasted Tomato Soup" 

aSelection[2] = "Cajun Seafood Bouillabaisse" 

aSelection[3] = "Pumpkin Crème Brulee" 

Using arrays in web applications can be useful for making code more readable. The ability to 

store multiple items in a single variable name can make code less confusing to read and 

reference. You can use array functions to calculate minimum, maximum, and averages on 

numeric values stored within an array, while character entries are easier to re-sort for 

alphabetical indexing.  

 

A value within an array is an element, the same as with lists. The position of an element in an 

array is its index and is indicated using bracket notation, as seen in the sample array above. 

The dimension of an array refers to its complexity. A simple array, like the one above, is a one-

dimensional array.  

 
Figure 1: A visual representation of a one-dimensional array 

A two-dimensional (2D) array is similar to a simple table.  

Figure 2: A traditional two-dimensional array 

A three-dimensional (3D) array is basically a cube of data, with rows, columns, and layers. 

Most developers will never use these types of arrays, but they are used for mapping 3-D 

graphics and can be useful when dealing with calendars. They are also heavily used in 

scientific and mathematical research and applications. 

 
Figure 3: One visualization of a 3D array 

 

Traditionally, arrays have a fixed size that stays constant and symmetrical with all rows having 

the same size. In ColdFusion, however, arrays are dynamically sized with rows of differing 

lengths based on the addition/removal of data. In essence, a two-dimensional array in 

ColdFusion is a one-dimension array whose elements contain other one-dimension arrays. 

 

 
Figure 4: A ColdFusion two-dimensional array with dynamically sized rows 

ColdFusion arrays differ from those in other languages in four ways: 

1. Its index starts with 1. In most languages, the first element in an array is index 0, but in 

ColdFusion, the first element is at index 1. 

2. Arrays are limited to three dimensions; in other languages arrays can contain as many 

as 32 dimensions and beyond 

3. Array sizes are not fixed, but change dynamically as data is added and removed 

4. When items are deleted from an array, the array is automatically compressed so that 

empty indexes are not left behind 

Creating and Displaying 

You can create an array in ColdFusion in two ways, implicitly with the ArrayNew() function and 

a <cfset> or explicitly using bracket notation. ArrayNew() takes a single argument, the 

dimension of the array (1-3). Once the array is created, you can then populate it using <cfset> 

statements and specifying the index using bracket notation. 

<cfset SelectionArray = Arraynew(1)> 

<cfset SelectionArray[1] = "Roasted Tomato Soup"> 

<cfset SelectionArray[2] = "Cajun Seafood Bouillabaisse"> 

<cfset SelectionArray[3] = "Pumpkin Crème Brulee"> 

The same array definition can also be rewritten to declare the array implicitly with bracket 

notation in a single line of code. 

<cfset SelectionArray = ["Roasted Tomato Soup", "Cajun Seafood 
Bouillabaisse", "Pumpkin Crème Brulee"]> 

When declaring arrays implicitly, notice that the element data starts and ends with brackets, 

with element data separated by commas and text values wrapped in quotes. In numeric arrays, 

quotes are optional. With implicit declarations, array elements are limited to simple values and 

the results of a CFML function. For arrays with complex values, you'll have to use an explicit 

declaration. 

 

To display the value of a specific array index, you again use the square bracket notation. 

<cfoutput>#Variables.SelectionArray[1]#</cfoutput> 

As with any variable, you can also view the contents of the entire array using the <cfdump> 

tag. 

<cfdump var="#Variables.SelectionArray#" label="Dinner Order" /> 

Array Functions 

As you would expect, ColdFusion has various built-in functions to use for basic array 

manipulation. For most of the functions, when you are working in a multi-dimensional array, 

you  must specify all but the last index in the array parameter.  

<cfset ArrayInsertAt(myarray[2], 4, "test")> 

Most of the functions will also throw an error if you attempt to reference an index that does not 

exist. 

ArrayAppend(array, value) 

Adds a new value to the end of an array 

ArrayAvg(array) 

Calculates the average of the values in an array; if the array is empty, it returns 

zero 

ArrayClear(array) 

Deletes all of the data in an array. 

ArrayDeleteAt(array, position) 

Deletes an element from a specified index; the indexes of the remaining 

elements are recalculated; so if you deleted an element at position five, the 

element that was at position six would now be at position five and so on 

ArrayInsertAt(array, position, value) 

Insert a value at a specified index; elements whose indexes are greater than or 

equal to the specified position have their indexes incremented by one; if the 

specified position is beyond the size of the array, an error occurs. 

ArrayIsDefined(array, index) 

Determines whether a specific index exists in an array; returns true/false 

ArrayIsEmpty(array) 

Checks to see if an array is empty; returns true/false 

ArrayLen(array) 

Determines the number of items in an array; in a one-dimensional array, this 

value will also equal the index of the last item in the array 

ArrayMax(array) 

Returns the largest numeric value in an array; if the array is empty, it returns 0 

ArrayMin(array) 

Returns the smallest numeric value in an array; if the array is empty, it returns 0 

ArrayNew(dimension) 

Creates a new array of 1-3 dimensions 

ArrayPrepend(array, value) 

Adds a new value to the beginning of an array 

ArrayResize(array, minimumsize) 

Resets an array to have a specified minimum number of elements; resizing an 

array to its expected number of elements can help improve performance; if 

working with an array with more than 500 elements, you should immediately use 

ArrayResize() after creating the array. 

ArraySet(array, start_pos, end_pos, value) 

Sets the elements in a specified index range to a single specified value; only for 

use with one-dimensional arrays; useful for initializing a new array 

ArraySort(array, sort type, [sort order]) 

Sorts an array using the specified type: numeric (sort numbers), text 

(alphabetical), or textnocase (case insensitive alphabetical); optional sort order 

can be asc (ascending) or desc (descending); not usable on non-simple values 

ArraySum(array) 

Sums the values in a specified array; if the array is empty it returns 0 

ArraySwap(array, position1, position2) 

Switch the values in an array at two specified positions. 

ArrayToList(array, [delimiter]) 

Converts a one-dimension array to a list 

IsArray(variable, [number]) 

Determines if a given variable is an array or not; you can add an optional 

dimension parameter to have the function test whether a variable is an array of a 

specific dimension 

ListToArray(list, [delimiter], [includeEmptyFields]) 

Convert a delimited list to a single-dimensional array; as array operations 

execute faster than list operations, converting lists to array can improve 

performance 

Looping Over Arrays 

Just as with lists and queries, you can use an index loop to loop over an array where the from 

and to specify the start and end points in the array and the index attribute is used to store the 

index you are currently looping over. You can combine this with the ArrayLen() function to 

specify the last array index and loop over an entire array. 

<cfset iLength = arraylen(SelectionArray)> 

<cfoutput> 

<cfloop from="1" to="#variables.iLength#" index="i"> 

#SelectionArray[i]#<br /> 

</cfloop> 

</cfoutput> 

Note that we stored the result of ArrayLen in a variable before using it in the loop. This also 

improves performance. In ColdFusion 8, you can also now loop over arrays directly in <cfloop> 

using the array attribute. 

<cfoutput> 

<cfloop array="#SelectionArray#" index="thisitem"> 

#variables.thisitem#<br /> 

</cfloop> 

</cfoutput> 

Part 2: Structures 
Structures are collections of related data, and the equivalent of associative arrays found in 

other programming languages. Like arrays, they store sets of information under a single 

variable name, however structures are indexed by name, also called a key, rather than 

numerical indexes, and the data is stored in key-value pairs. 

 
Figure 5: A visual representation of a structure 

Structures do not actually store "keys" but rather it associates a value with another value. The 

use of associated keys and values make structures very useful for tasks such as: 

 Categorizing information in descriptively named variable groups 

 Searching for values based on keys 

 Outputting keys and values together 

Creating and Displaying 

As with arrays, structures can be created explicitly with a function, or implicitly using either 

bracket notation (similar to arrays) or dot notation. 

<cfset stDir = StructNew()> 

You can also create a structure by simply populating one. The first time you set a value in a 

non-existent structure, ColdFusion will first create the structure for you. Once created, a 

structure can be populated with bracket notation (array syntax), dot notation, or any 

combination there in. When using array syntax, the structure's literal key names are enclosed 

in quotation marks; otherwise, the name would be evaluated as a variable. 

<cfset stDir["Name"] = "Menu"> 

 

Array syntax allows you to use keys that contain special characters, spaces, and variables; it 

also respects cases in keys. However, as ColdFusion variable names should not contain 

spaces or special characters, it is generally better not to use them for your structure keys. 

 

Dot notation is common in object-oriented languages like Java in which an option property is 

referred to using the syntax 

object.property 

For structures, the syntax is similar 

Strcturename.keyname 

 

When using dot notation, the key name cannot contain special characters or spaces, and you 

have to use special notation to use variables. The key names are stored in uppercase. Dot 

notation is typically used with associating descriptive elements to a certain type, for example: 

<cfset stDir.name="Menu"> 

<cfset stDir.desc="Daily Menus"> 

 

In ColdFusion 8, you can also create structures implicitly using curly-braces wrapping comma-

separated key name/value pairs: 

<cfset stDir = { 

url = "/about", 

name = "About Cafe Townsend", 

id="4" 

}> 

When using this method, you cannot populate nested structures nor can you include dots in 

key names. Again, the key names will be stored in uppercase. 

 

As noted above, with bracket notation, you can control the case of your key names. This can 

be important if you want to display the key themselves in your output and are concerned about 

how it looks. If you use bracket notation, whatever case you use to set the key will be the case 

used for storing it. In other methods, keys are automatically converted to all upper case. 

Therefore, if we ran this code then used <cfdump> to output it: 

<cfset stDir=StructNew()> 

<cfset stDir.url="/about/"> 

<cfset stDir["Name"]="About Café Townsend"> 

<cfset stDir["ID"]="4"> 

 
Figure 6: Only the keys declared using bracket notation retain their case 

Note that this has no effect on your ability to refer to the variables (which is case insensitive), 

but only refers to potential issues in outputting the keys. 

 

Similarly, you can use either dot or bracket notation to print the values in a structure. In bracket 

notation, use the key as the "index" value, while with dot notation you use structure.key 

notation. 

#stDir["name"]# 

#stDir.name# 

Structure Functions 

Just as with arrays, you have a variety of functions available for manipulating structures. 

Duplicate(variable) 

Creates a perfect clone of a variable, with no reference to the original variable left 

behind; used to duplicate complex structures, such as nested structures and 

queries 

IsStruct(variable) 

Determines if a given variable is a valid ColdFusion structure, or a Java object 

that implements the java.lang.Map interface; returns true/false 

StructAppend(structure1, structure2, [overwriteflag]) 

Appends one structure to the end of another; set overwriteflag to yes/true to have 

any values in struct2 overwrite any corresponding ones in struct1 (defaults to no) 

StructClear(structure) 

Removes all keys and values from a specified structure, but does not remove the 

structure itself 

StructCopy(structure) 

Copies a structure, copying all top-level keys, values and arrays in the structure 

by value; copies of nested structures are copied by reference, so if the nested 

structure is changed, the copy will change as well 

StructCount(structure) 

Counts the number of top-level keys in a structure 

StructDelete(structure, key, [indicatenotexisting]) 

Removes an element from a structure as specified by a key; returns true unless 

you use the optional flag of indicatenotexisting to have it return false if the key 

doesn't exist; by default indicatenotexisting is set to false 

StructFind(structure, key) 

Determines the value associated with a key in a structure 

StructFindKey(top, value, scope) 

Recursively searches through structures (or nested arrays), to find elements in a 

structure whose key matches the search value; use scope to have the function 

find the first instance ("one") or all instances ("all"). The function returns an array 

that includes one structure for each element found containing: owner - the parent 

object containing the key, path - the path to the key relative to the top, and value 

- the value of the key 

StructFindValue(structure, value, scope) 

Recursively searches through structures (or nested arrays), to find keys whose 

value exactly matches the search key; use scope to have the function find the 

first instance ("one") or all instances ("all"). The function returns an array that 

includes one structure for each value found containing: owner - the parent object 

containing the key, path - the path to the key relative to the top, and value - the 

value of the key 

StructGet(pathDesired) 

Retrieve a structure from a specified path; if the structure is not found, it will be 

created 

StructInsert(structure, key, value, [allowoverwrite]) 

Inserts a key/value pair into a structure with option to overwrite an existing like-

named key; when using StructInsert the key retains its case 

StructIsEmpty(structure) 

Indicates is a structure is empty (true) or not (false) 

StructKeyArray(structure) 

Finds the keys in a structure and returns them in an array 

StructKeyExists(structure, "key") 

Determines if a specified key exists in a structure 

StructKeyList(structure) 

Finds the keys in a structure and returns them in a list 

StructNew(structure) 

Creates an empty structure 

StructSort(base, [sortType], [sortOrder], [pathToSubElement]) 

Returns a sorted array of the top level keys in a structure; numeric (sort 

numbers), text (alphabetical; default), or textnocase (case insensitive 

alphabetical); optional sort order can be asc (ascending; default) or desc 

(descending) and optional pathToSubElement indicates which element in the 

structure should be used to sort the structure, by default it will sort on the top 

level; note that function does not change the underlying structure itself 

StructUpdate(structure, key, value) 

Updates a key with a specified value 

Looping Over Structures 

To loop over a structure in ColdFusion, you use a collection loop, with the collection attribute 

specifying the name of the structure and the item attribute giving the name of the variable to 

use inside the loop to revert to the name of the current key. 

<!--- Create and fill the structure ---> 

<cfset stDir= { 

ID="1", 

Name="Cafe Townsend", 

Desc="Cafe Townsend Root", 

URL="/" 

}> 

<!--- Use a collection loop ---> 

<cfloop collection="#stdir#" item="key"> 

<cfoutput>#key#<br /></cfoutput> 

</cfloop> 

The output would look like this: 

ID 

DESC 

URL 

NAME 

You can also have the loop output the key and its value (provided the value is simple and not a 

nested structure): 

<cfloop collection="#stDir#" item="key"> 

<cfoutput>#key#: #stDir[key]#<br /></cfoutput> 

</cfloop> 

which would give us: 

ID: 1 

DESC: Cafe Townsend Root 

URL: / 

NAME: Cafe Townsend 

Part 3: Complex Data and Complex Data Structures 

Accessing Complex Data 

In ColdFusion, as with many programming languages, you can reference and pass complex 

data types by value, in which a physical copy is made of the data, or by reference, when one 

variable points to another (aliases). Like simple variables, arrays are usually accessed by 

value. So if you use a <cfset> to create a copy of an array, then a new physical copy is created 

and they are independent entities. Changes made in one array will have no effect on the other.  

 
Figure 7: When copying an array with <cfset>, a duplicate is made that is independent of the original 

However, structures and queries are accessed by reference. When you use a <cfset> to create 

a copy of the structure, you really are creating an alias for the structure. Only one copy of the 

data exists, with the new name being a pointer to the original structures values, and any 

changes made to one structure is reflected in both. We discussed this briefly when we looked 

at the difference between Duplicate() and StructCopy() earlier. 

 
Figure 8: When copying a structure or query, both names point to the same single set of data 

If you need to create completely independent copies of complex variables, you will need to use 

the Duplicate() function, which can be used on any data type, and will make independent 

copies of all of the data at all levels of a complex variable. This is referred to as a "deep copy."   

 
Figure 9: Using Duplicate(), we can create independent copies of a structure 

Remember when we discussed StructCopy(), we noted that it makes a "shallow copy" in that it 

only makes a physical copy of the top level of the structure, while the embedded structures are 

only copied as pointers.  

 
Figure 10: With StructCopy(), only the top level is independent, while the substructures remain references 

If you change one of the key values of the top-level structures, the value will not carry over to 

the copy. However, if you change one of the key values of the embedded structure, it will 

change in both. For example: 

<cfset stNew.name="Cheesecake"> 

<cfset stNew.nutrition.calories ="1100"> 

<cfset stNew.nutrition.fatgrams = "37"> 

Figure 11: Notice the names are difference, but the change in calories and fat carried through to both 

structures 

Dynamic Variables 

Values sent through a form are stored in a structure called FORM, which you can access using 

structure dot notation. You can also access ColdFusion's other variable scopes, such as 

Application, Session, CGI, etc, using structure dot notation. With forms, however, you often 

have variable names that are determined dynamically at runtime. You can access the names 

of these variables programmatically by looping over the form structure and displaying the index 

of the loop as indicated by the following example 

<cfoutput> 

<cfloop collection="#form#" item="i"> 

#i#<br> 

</cfloop> 

</cfoutput> 

The structure dot notation, you can access the values of the dynamic form variables.  

<cfoutput> 

<cfloop collection="#form#" item="i"> 

#i#: #form[i]#<br> 

</cfloop> 

</cfoutput> 

You can also use the Evaluate() function to perform the same task, however this method slows 

performance and is strongly discouraged as dot structure notation is faster, particularly with 

large loops, and cleaner syntactically. 

<cfoutput> 

<cfloop collection="#form#" item="i">  

#i#: #Evaluate("form." & i)#<br> 

</cfloop> 

</cfoutput> 

Combining Complex Data Types 

The elements in arrays and structures are not limited to simple values; they can be complex 

data types. You can have an array of structures, an array of queries, etc. Arrays, structures, 

and queries can be combined into flexible, complex data structures. Let's say we had a 

restaurant management system in which you need to keep track of patron orders. You'll want 

to track the name, quantity, and price of each item. You can use a basic structure like this: 

 
Figure 12: Structure of a Restaurant Menu Order 

However if you need to determine and manipulate the number of items in a shopping cart like 

this, you'll find it difficult using this method. The number of ordered items needs to remain 

flexible, so you can use an array to store the individual structures: 

<cfset aCart=ArrayNew(1)> 

<cfset aCart[1]= { 

quantity = "4", 

name = "Roasted Tomato Soup", 

price = "12.00" 

}> 

<cfset aCart[2]= { 

quantity="1", 

name="Cajun Seafood Bouillabaisse", 

price="13.50" 

}> 

 
Figure 13: <cfdump> of an array of structures 

Of course, when you create complex data structures that combine arrays, queries, and/or 

structures, you will have to use a combination of syntax to access their values. For example, 

display the value of the name key in the structure occupying the second array element in our 

array of structures above, you can use a combination of array and structure notation, or just 

bracket notation: 

<cfoutput>#aCart[2].name#</cfoutput> 

<cfoutput>#aCart[2]["name"]#</cfoutput> 

With an array of arrays, you would use bracket notation by itself, with the first bracket being the 

index of the main array, the second being the child, and so on. For example, in this array of 

arrays: 

 
Figure 14: <cfdump> from an array of arrays 

We would display the second element, "Roasted Tomato Soup", using: 

<cfoutput>#aOrderData[1][2]#</cfoutput> 

You can also put queries inside other data structures. You have actually seen this in action 

when you use a <cfdump> tag on query sets! The query is actually put inside a structure that 

also contains the query in a key called ResultSet, along with the Cached, ExecutionTime, and 

SQL keys. If we put a query inside an array, you would access the query using bracket 

notation. In this array of a query: 

Figure 15: A partial <cfdump> of an array of queries 

We could access the value of the "dishname" field in the fifth row of the query, "Crispy 

Flatbread", using: 

<cfoutput>#aDishes[1].dishname[5]#</cfoutput> 

Best Practices and Reminders 
1. ColdFusion arrays are different from typical arrays in that its index starts at 1, it is 

limited to three-dimensional arrays, and its arrays are dynamically sized 

2. Arrays can be declared explicitly with ArrayNew() or implicitly using bracket notation 

3. If you know the expected size of an array, you can improve performance by using 
ArrayResize() after creating it to expand an array to the expected number of elements 

4. Convert lists to arrays for faster performance 

5. When using the results of ArrayLen() in other operations, store the value in a variable 
first 

6. When populating arrays, use bracket notation (array syntax) if retaining the case of the 
key name is important or if you want to use special characters, spaces, or variables as 
key names 

7. Structures are collections of related data stored under a single variable name and 
indexed by key names rather than numbered indexes 

8. StructFindKey() will find matching keys in a structure, while StructFindValue() finds 
matching values 

9. Use dot notation over Evaluate() to access the values of dynamically named variables 
whenever possible for faster performance 

10. Complex data types can be combined into complex data structures, such as arrays of 
structures, structures of queries, etc 

11. Structures and queries are accessed by reference, while simple variables and arrays 
are accessed by value. 

12. Use Duplicate() to create deep copies of complex data including structures and queries; 
remember the StructCopy() function will only make an independent copy of the top level 
structure, so that embedded structures will still be a single entity and changes in one will 
change the other. 

Applying What We Learned 
1. Practice working with arrays 

a. Create an array explicitly and implicitly 
b. Populate it 
c. Use the various array functions to add elements, remove elements, order the 

elements 
d. Perform the various mathematical functions on a numerical array 
e. Output an array using a <cfdump> and a <cfloop> 
f. Try making a 3D array and outputting it with <cfdump> 

2. Practice working with structures 
a. Create a structure explicitly and implicitly with nested elements 
b. Test whether a structure exists 
c. Dump a query result set into a structure 
d. Use the various structure functions to delete a key and to clear the structure 

3. Use dot notation to refer to a value in a dynamically named variable 
4. Experiment with copying arrays, structures, and queries, using <cfset> then using 

Duplicate() and study the effects of changing data in both instances 
5. Experiment with complex data structures 

a. Make an array of structures 
b. Make a structure of queries 
c. Make an array of queries and structures 

